

# Ein Ausflug mit SAS in die HTML 5 Welt

Frank Biedermann  
Grünenthal GmbH  
Zieglerstraße 6  
52078 Aachen

[Frank.Biedermann@grunenthal.com](mailto:Frank.Biedermann@grunenthal.com)

## Zusammenfassung

Zusammen mit SAS 9.4 wurde eine Vielzahl von neuen ODS Ausgabekanälen bereitgestellt, einer davon ist HMTL5. Das ODS HMTL5 Statement unterscheidet sich von ODS HTML im Wesentlichen darin, dass es HTML Version 5.0 anstatt HTML Version 4.01 verwendet. Dadurch bekommt ein SAS Benutzer die Möglichkeit dynamische HTML5 Ausgabedateien zu erstellen. Des Weiteren verwendet ODS HTML5 als Standard Ausgabeformat für Grafiken das Scalable Vector Graphics Format (SVG) anstatt dem alten (PNG) Portable Network Graphics Format. In diesem Beitrag werden die Unterschiede zwischen ODS HTML und ODS HMTL5 mit dem Fokus auf Grafiken gezeigt, des Weiteren demonstriere ich, wie mit einigen ODS Tricks die Reports optimiert und dynamisch gestaltet werden können und wie mit ODS HTML5 Audio und Video Inhalte in Reports eingebunden werden können.

**Schlüsselwörter:** ODS HTML, ODS HTML5, PNG, CGI, SVG, CSS

## 1 Was ist HTML 5?

HTML 5 ist eine Computersprache zur Auszeichnung und Vernetzung (Auszeichnungssprache) von Texten, Bildern, Videos und weiteren Elementen in einem Webdokument. Die Inhalte können plattformübergreifend dargestellt werden. HTML 5 ersetzt stufenweise die HTML 4 Spezifikation.

HTML 5 beinhaltet Code-Statements um dynamische Inhalte ohne die Einbindung externer Plugins wiederzugeben. Des Weiteren können mit den neuen Tags und Attributen Webanwendungen für die verschiedensten Plattformen wie Smartphone, Tablet oder PC entwickelt werden. Der Inhalt kann besser strukturiert und ausgezeichnet werden.

## 2 Was ist mit SAS 9.4 M6 möglich um HTML 5 einzusetzen?

**Tabelle 1:** Übersicht SAS Display Manager, SAS EG, SAS Studio

Eigenschaften	SAS Display Manager	SAS Enterprise Guide	SAS Studio
Ausgabe (Default)	HTML 4	HTML 4	HTML 5
Erstellen	Ja	Ja	Ja
Scalable Vector Graphics (SVG)	Ja	Ja	Ja

Musik	Ja	Ja	Ja
Video	Ja	Ja	Ja

ODS HTML5 ist seit Version 9.4 verfügbar

Die wichtigste Änderung die mit ODS HTML 5 eingeführt wurde, ist die Ausgabe von Grafiken. Das Standard Grafik Ausgabeformat für ODS HTML ist das *Portable Network Graphics (.png)* Format. In ODS HTML 5 ist das Standard Grafik Ausgabeformat *Scalable Vector Graphics (svg)*. Durch diese Änderung wird die Grafikqualität von Plots erheblich verbessert.

### 3 Einführung in ODS HTML 5

HTML 5 ersetzt nicht ODS HTML, welches auf der HTML 4.01 Spezifikation aufsetzt. ODS HTML ist weiter verfügbar, muss aber über einen separaten ODS Kanal geöffnet werden.

Beispiel Code ODS HTML:

```
filename ausgabe 'C:\temp\report1.html';
ods html body=ausgabe style=HTMLBlue;
title1 'HTML Report 1';
footnote;
proc ttest;
  class Gender;
  var Score;
run;
ods html close;
```

ODS HTML Ausgabe  
Datei




Abbildung 1: ODS HTML

## Beispiel Code ODS HTML5:

```
filename ausgabe 'C:\temp\report1.html';
ods html5 body=ausgabe style=HTMLBlue;
title1 'HTML5 Report 1';
footnote;
proc ttest;
  class Gender;
  var Score;
run;
ods html5 close;
```


ODS HTML5 Ausgabe  
Datei

Abbildung 2: ODS HTML5

ODS HTML5 erzeugt automatisch einen Mouse Over Effekt in diesem Beispiel „Summary Panel for Score“.

## 4 ODS HTML 5 Beispiele

### 4.1 PROC ODSTEXT

In diesem Beispiel wird mit der Hilfe von PROC ODSTEXT ein HTML5 Dokument mit Hyperlinks gebaut, diese Hyperlinks dienen als Sprungmarken.

Beispiel Code ODS HTML5 – PROC ODSSTEXT:

```
data a;
  set sashelp.demographics (where=(region='EUR'));
  fmtpop = pop;
run;
proc sort;
  by pop;
run;
proc format;
  value population
  low - <1000000 = klein
  1000000 - <10000000 = Mittel
  10000000 - <100000000 = Groß
  100000000 - <1000000000 = Sehr groß
;
run;

ods html5 (id=web) style=daisy path='C:\temp'
          file='report2.html'
          (title='Erstelle Links zu den
Ausgabebetabellen')
          options(outline='no')
          anchor='meinreport';

title;
footnote;
proc odstext contents='';
  p "<h1>Europas Bevölkerung von klein bis sehr groß</h1>";
  p "<h2><a href='#myreport1'>Keine Bevölkerung</a><h2>";
  p "<h2><a href='#myreport2'>Mittlere Bevölkerung</a><h2>";
  p "<h2><a href='#myreport3'>Große Bevölkerung</a><h2>";
  p "<h2><a href='#myreport4'>Sehr große Bevölkerung</a><h2>";
run;

proc print data=a label;
  var name pop gni;
  id name;
  format fmtpop population.;
  by fmtpop;
run;
ods html5 close;
```

ODS HTML5 – PROC  
ODSTEXT Ausgabe  
Datei

### Europas Bevölkerung von klein bis sehr groß

[Keine Bevölkerung](#)

[Mittlere Bevölkerung](#)

[Große Bevölkerung](#)

[Sehr große Bevölkerung](#)

---

fmtpop=klein

GLC Country Name	GLC Country Name	Population (2005)	Gross National Income per Capita (PPP Int.\$ 2004)
SAN MARINO	SAN MARINO	28,117	..
MONACO	MONACO	35,253	..
ANDORRA	ANDORRA	67,151	..
ICELAND	ICELAND	294,561	32360
MALTA	MALTA	401,630	18720
LUXEMBOURG	LUXEMBOURG	464,904	61220
CYPRUS	CYPRUS	835,307	22330

fmtpop=Mittel

GLC Country Name	GLC Country Name	Population (2005)	Gross National Income per Capita (PPP Int.\$ 2004)
ESTONIA	ESTONIA	1,329,697	13190
SLOVENIA	SLOVENIA	1,966,814	20730
MACEDONIA	MACEDONIA	2,034,060	6480
LATVIA	LATVIA	2,306,988	11850
ARMENIA	ARMENIA	3,016,312	4270
ALBANIA	ALBANIA	3,127,778	..

**Abbildung 3: ODS HTML5 – PROC ODSTEXT**

## 4.2 Report Writing Interface (RWI) Optionen

In dem nächsten Beispiel wird gezeigt wie man mit dem Report Writing Interface kurz RWI Optionen ein HTML 5 Report mit Bildern erstellen kann.

Beispiel Code ODS HTML5 – RWI Optionen:

```
%let logopath=C:\Temp\logo;

ods html5 path="C:\Temp" file="report3.html" (title="KSFE e.V.");

title; footnote;

data _null_;
  dcl odsout obj();
  obj.image(file: "&logopath./ksfe-ev.png",
            description: "KSFE e.V.");
run;

data _null_;
  dcl odsout obj();
  obj.note(data: "Unsere Sponsoren");
run;
```

## *F. Biedermann*

```
title; footnote;
ods layout gridded columns=10 rows=1 row_gutter=0 column_gutter=0;
ods region;
data _null_;
  dcl odsout obj();
  obj.image(file: "&logopath./sas.jpg",
    description: "SAS Germany");
run;

ods region;
data _null_;
  dcl odsout obj();
  obj.image(file: "&logopath./clinstat_logo.png", width: "170px",
height: "90px",
  description: "Clinstat Logo");
run;

ods region;
data _null_;
  dcl odsout obj();
  obj.image(file: "&logopath./hms.jpg",
    description: "HMS");
run;

ods region;
data _null_;
  dcl odsout obj();
  obj.image(file: "&logopath./icasus.jpg",
    description: "Icasus");
run;

ods region;
data _null_;
  dcl odsout obj();
  obj.image(file: "&logopath./parexel_logo.jpg", width: "200px",
height: "70px",
  description: "Parexel");
run;

ods region;
data _null_;
  dcl odsout obj();
  obj.image(file: "&logopath./viadee_logo_rot.png", width:
"200px", height: "70px",
  description: "Viadee");
run;

ods layout end;
ods html5 close;
```



**Abbildung 4:** ODS HTML5 – RWI

### 4.3 ODS Layout – Grid

In diesem Beispiel wird mit der Hilfe von ODS LAYOUT ein HTML5 Dokument mit zwei Tabellen im Grid Style nebeneinander erstellt.

Beispiel Code ODS HTML5 – ODS LAYOUT:

```
ods html5 path="C:\Temp" file="report4.html" style=HTMLBLUE;
ods layout gridded columns=2 column_gutter=1in;
```

```
title; footnote;
ods region;
proc print data=sashelp.class(obs=5);
run;
```

```
ods noproctitle;
ods region;
proc means data=sashelp.class n mean;
var height weight;
run;
```

```
ods layout end;
ods html5 close;
```

Obs	Name	Sex	Age	Height	Weight
1	Alfred	M	14	69.0	112.5
2	Alice	F	13	56.5	84.0
3	Barbara	F	13	65.3	98.0
4	Carol	F	14	62.8	102.5
5	Henry	M	14	63.5	102.5

Variable	N	Mean
Height	19	62.3368421
Weight	19	100.0263158

**Abbildung 5:** ODS HTML5 – ODS LAYOUT

## 4.4 Cascading Style Sheets

Im nächsten ODS HTML5 Report Beispiel, wird für die Layout Gestaltung ein Cascading Style Sheet (CSS) eingebunden.

Beispiel Code ODS HTML5 – CSS und Styles:

```
ods html5 path="C:\Temp" file="report5.html" style=minimal
  headtext='<style>td {color: green; font-style: italic;}</style>';

title; footnote;

proc print data=sashelp.class (obs=10) noobs
  style(table)={prehtml="<p style='text-align: center;
    font-weight: bold;
    color: blue;'>Class</p>"}
  style(data)={htmlstyle="font-weight: bold;"};
run;

ods html5 close;
```

Beispiel Code ODS HTML5 – Externe CSS Datei:

```
ods html5 path="C:\Temp" file="report5_css_extern.html"
  stylesheet=(URL="https://www.w3schools.com/w3css/4/w3.css");
```

ODS HTML5 –  
CSS Ausgabe  
Datei

Class				
Name	Sex	Age	Height	Weight
<i>Alfred</i>	<i>M</i>	<i>14</i>	<i>69.0</i>	<i>112.5</i>
<i>Alice</i>	<i>F</i>	<i>13</i>	<i>56.5</i>	<i>84.0</i>
<i>Barbara</i>	<i>F</i>	<i>13</i>	<i>65.3</i>	<i>98.0</i>
<i>Carol</i>	<i>F</i>	<i>14</i>	<i>62.8</i>	<i>102.5</i>
<i>Henry</i>	<i>M</i>	<i>14</i>	<i>63.5</i>	<i>102.5</i>
<i>James</i>	<i>M</i>	<i>12</i>	<i>57.3</i>	<i>83.0</i>
<i>Jane</i>	<i>F</i>	<i>12</i>	<i>59.8</i>	<i>84.5</i>
<i>Janet</i>	<i>F</i>	<i>15</i>	<i>62.5</i>	<i>112.5</i>
<i>Jeffrey</i>	<i>M</i>	<i>13</i>	<i>62.5</i>	<i>84.0</i>
<i>John</i>	<i>M</i>	<i>12</i>	<i>59.0</i>	<i>99.5</i>

Abbildung 6: ODS HTML5 – CSS

## 5 Standard ODS Grafik Ausgabeformate

**Tabelle 1:** Übersicht Grafik Ausgabeformate bei ODS Ausgabekanal

ODS Ausgabe Kanal	Standard Ausgabeformat
HTML	PNG
HTML 5	SVG
LISTING	PNG
RTF	PNG
PCL	SVG
PDF	SVG
POSTSCRIPT (PS)	PNG
PRINTER	Standard Client Printer

## 6 Plots und Grafiken

ODS HTML 5 verwendet defaultmäßig Scalable Vector Graphics (SVG). Sollen Grafiken in einem anderen Format als SVG erstellt werden wird folgendes Code Statement benötigt: `ods graphics on / imagefmt=png;`

Beispiel Code ODS HTML 5 – SVG/PNG



```
data my_data;
  set sashelp.us_data
    (where =(statecode in ('PA' 'WV' 'VA' 'NC' 'SC' 'GA' 'FL')));
run;
ods html5 path="C:\Temp" file="report6.html";

ods graphics on / imagefmt=png;

title; footnote;

axis1 label=none value=(justify=right);
axis2 label=('2012 Population') value=(font="albany amt") minor=none
offset=(0,0);
pattern1 v=s c=cx499DF5;

proc gchart data=my_data;
  hbar statename / type=sum sumvar=population_2010 descending
  nostats noframe
  maxis=axis1 raxis=axis2
  autoref clipref cref=graydd
  html=my_html des='';
run;
quit;
ODS HTML5 CLOSE;
ODS LISTING;
```



ODS HTML 5 bietet auch die Möglichkeit an zu wählen wie eine Grafik verknüpft werden soll. Als eigene Datei oder als Teil der HTML 5 Ausgabedatei. Es gibt zwei Optionen: `options(bitmap_mode=)` für Rastergrafik Formate und `options(svg_mode=)` für SVG Grafiken.

**Mögliche Parameter**

EMBED = erstellt einen HTML5 `<embed>` tag

`<embed src="c:\Temp\gchart2.png"/>`

IMG = erstellt einen HTML5 `<img>` tag der die Grafik anzeigt

``

INLINE = fügt die Grafik in das HTML5 ein

LINK = erstellt einen Hyperlink der auf die

Grafikdatei verweist

`<a href=" c:\Temp\gchart2.png" GChart</a>`

OBJECT erstellt einen HTML5 `<object>` tag der die Grafik anzeigt

`<object data=" c:\Temp\gchart2.png" </object>`

## Beispiel Code ODS HTML 5 – SVG/PNG

```
ods html5 path="C:\Temp" file="report7.html"
  options(bitmap_mode='inline');
ods graphics on / imagefmt=png;
title; footnote;
axis1 label=none value=(justify=right);
axis2 label=('2012 Population') value=(font="albany amt") minor=none
offset=(0,0);
pattern1 v=s c=cx499DF5;
proc gchart data=my_data;
  hbar statename / type=sum sumvar=population_2010 descending
  nostats noframe
  maxis=axis1 raxis=axis2
  autoref clipref cref=graydd
  html=my_html des='';
run;
quit;
ODS HTML5 CLOSE;
```

## 7 Animierte Grafiken

ODS HTML 5 bietet ähnlich wie HTML die Möglichkeit animierte SVG Grafiken und Plots anzuzeigen.

### Beispiel Code ODS HTML 5 – Animierte Grafiken

```
options animation=start animduration=3 noanimoverlay
  nodate nonumber;

ods html5 path="C:\Temp" file="report8.html";


  data work.qsales;
    set sashelp.prdsale (keep=Actual Country Product Quarter
Year);
  run;

  proc sort data=work.qsales;
    by quarter;
  run;

  axis1 order=(0 to 25000 by 5000);

  title1 "1993 Sales";
  proc gchart data=work.qsales (where=(year=1993));
    vbar3d country / sumvar=actual subgroup=product sum
      raxis=axis1 shape=cylinder name='sales93';
    where product in ("BED" "TABLE" "CHAIR");
    by quarter;
```

```
run;  
quit;  
  
title1 "1994 Sales";  
proc gchart data=work.qsales(where=(year=1994));  
  vbar3d country / sumvar=actual subgroup=product sum  
    raxis=axis1 shape=cylinder name='sales94';  
  where product in ("BED" "TABLE" "CHAIR");  
  by quarter;  
run;  
quit;  
  
ods html5 close;  
options animation=stop;
```



## 8 Audioinhalte einbinden

Mit ODS HTML5 ist es sehr einfach geworden Audiodateien in die Ausgabe einzubauen. Der Befehl `<AUDIO>` im HTML5 Code ermöglicht dieses. Jeder moderne Web Browser (Chrome, Firefox, Edge, Safari) unterstützt diesen Befehl. Das Einbinden eines externen Plugins oder Scripts ist nicht mehr nötig. Dies ermöglicht eine nahtlose Integration von Audioinhalten in einem Report. Die Methode AUDIO ist Teil des Report Write Interfaces, sie kann nur mit ODS HTML5, ODS EPUB oder ODS EPUB3 verwendet werden.

Beispiel Code ODS HTML 5 – Audioinhalte

```
ods html5 path="C:\Temp" file="report9.html";
title1 "Beispiel: Daft Punk's Da Funk";
data _null_;
dcl odsout obj();
obj.audio(
file:"https://upload.wikimedia.org/wikipedia/en/3/3d/Sample_of_Daft_
Punk%27s_Da_Funk.ogg",
  type: "ogg",
  preload: "auto",
  autoplay: "yes", /* on, off or no, yes */
  loop: "yes" /* on, off or no, yes */
  /* muted: on, off or no, yes */
);
run;
ods html5 close;
```



**Tabelle 2:** Übersicht unterstützte Audio Formate bei Web-Browser

Browser	Audio Formate
Internet Explorer	MP3
EDGE	MP3
Firefox	MP3, Wav, OGG
Chrome	MP3, Wav, OGG
Opera	MP3, Wav, OGG
Safari	MP3, Wav

**Tabelle 3:** Übersicht unterstützte Audio Formate bei der SAS RWI AUDIO Funktion

TYPE: "audio-type"
MP3
Wav
OGG

## 9 Videoinhalte einbinden

ODS HTML 5 hat auch das Einbinden von Videoinhalte in Ausgabedateien vereinfacht. Die Methode VIDEO ist Teil des Report Write Interfaces, sie kann nur mit ODS HTML5 verwendet werden.

### Beispiel Code ODS HTML 5 – Videoinhalte

```
ods html5 path="C:\Temp" file="report10.html";
data _null_;
dcl odsout obj();
obj.video(file:
"https://upload.wikimedia.org/wikipedia/commons/d/d7/STS-114-launch-
Jul26-2005.ogv",
    type: "ogg",
    width: "640",
    height: "480",
    preload: "auto",
    autoplay: "yes",
    poster:
"https://upload.wikimedia.org/wikipedia/commons/0/07/Sts115-s-
040.jpg",
    loop: "on"
);
run;
ods html5 close;
```

ODS HTML5 –  
Videoinhalte Ausgabe  
Datei Browser



Abbildung 11: ODS HTML5 – Video

**Tabelle 4:** Übersicht unterstützte Video Formate bei Web-Browser

Browser	Video Formate
Internet Explorer	MP4
EDGE	MP4
Firefox	MP4, WebM, OGG
Chrome	MP4, WebM, OGG
Opera	MP4, WebM, OGG
Safari	MP4

**Tabelle 5:** Übersicht unterstützte Video Formate bei der SAS RWI VIDEO Funktion

TYPE: "video-type"
MP4
WebM
OGG

## Literatur

- [1] ODS HTML 5 (SAS Support Page)  
<https://documentation.sas.com/?docsetId=odsug&docsetTarget=p0hcv8gpxqebnpn1is52we2enltx.htm&docsetVersion=9.4&locale=en>
- [2] Report Writing Interface (SAS Support Page).  
<https://documentation.sas.com/?docsetId=odsadvug&docsetTarget=n1a6s17kcol07zn1jsxn170bnj8k.htm&docsetVersion=9.4&locale=en>
- [3] Brandon George, Nathan Bernicchi und Paul Egeler 2018, Take a Dive into HTML5. <https://www.sas.com/content/dam/SAS/support/en/sas-global-forum-proceedings/2018/2670-2018.pdf>